

NAG Fortran Library Routine Document

F08GAF (DSPEV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08GAF (DSPEV) computes all the eigenvalues and, optionally, eigenvectors of a real n by n symmetric matrix A in packed storage.

2 Specification

```
SUBROUTINE F08GAF (JOBZ, UPLO, N, AP, W, Z, LDZ, WORK, INFO)
INTEGER          N, LDZ, INFO
double precision AP(*), W(*), Z(LDZ,*), WORK(*)
CHARACTER*1      JOBZ, UPLO
```

The routine may be called by its LAPACK name ***dspev***.

3 Description

The symmetric matrix A is first reduced to tridiagonal form, using orthogonal similarity transformations, and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: JOBZ – CHARACTER*1 *Input*
On entry: if JOBZ = 'N', compute eigenvalues only.
 If JOBZ = 'V', compute eigenvalues and eigenvectors.
Constraint: JOBZ = 'N' or 'V'.
- 2: UPLO – CHARACTER*1 *Input*
On entry: if UPLO = 'U', the upper triangle of A is stored.
 If UPLO = 'L', the lower triangle of A is stored.
- 3: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

- 4: $AP(*)$ – **double precision** array *Input/Output*

Note: the dimension of the array AP must be at least $\max(N \times (N + 1)/2)$.

On entry: the upper or lower triangle of the symmetric matrix A , packed columnwise in a linear array. The j th column of A is stored in the array AP as follows:

if $UPLO = 'U'$, $AP(i + (j - 1) \times j/2) = a_{ij}$ for $1 \leq i \leq j$;
 if $UPLO = 'L'$, $AP(i + (j - 1) \times (2 \times n - j)/2) = a_{ij}$ for $j \leq i \leq n$.

On exit: is overwritten by values generated during the reduction to tridiagonal form. If $UPLO = 'U'$, the diagonal and first super-diagonal of the tridiagonal matrix T overwrite the corresponding elements of A , and if $UPLO = 'L'$, the diagonal and first sub-diagonal of T overwrite the corresponding elements of A .

- 5: $W(*)$ – **double precision** array *Output*

Note: the dimension of the array W must be at least $\max(1, N)$.

On exit: if $INFO = 0$, the eigenvalues in ascending order.

- 6: $Z(LDZ,*)$ – **double precision** array *Output*

Note: the second dimension of the array Z must be at least $\max(1, N)$.

On exit: if $JOBZ = 'V'$, then if $INFO = 0$, Z contains the orthonormal eigenvectors of the matrix A , with the i th column of Z holding the eigenvector associated with $W(i)$.

If $JOBZ = 'N'$, Z is not referenced.

- 7: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08GAF (DSPEV) is called.

Constraints:

if $JOBZ = 'V'$, $LDZ \geq \max(1, N)$;
 $LDZ \geq 1$ otherwise.

- 8: $WORK(*)$ – **double precision** array *Workspace*

Note: the dimension of the array $WORK$ must be at least $\max(1, 3 \times N)$.

- 9: $INFO$ – INTEGER *Output*

On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$INFO < 0$

If $INFO = -i$, the i th argument had an illegal value.

$INFO > 0$

If $INFO = i$, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating point operations is proportional to n^3 .

The complex analogue of this routine is F08GNF (ZHPEV).

9 Example

To find all the eigenvalues of the symmetric matrix

$$A = \begin{pmatrix} 1 & 2 & 3 & 4 \\ 2 & 2 & 3 & 4 \\ 3 & 3 & 3 & 4 \\ 4 & 4 & 4 & 4 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08GAF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=10)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*      .. Local Scalars ..
      DOUBLE PRECISION EERRBD, EPS
      INTEGER          I, INFO, J, N
*      .. Local Arrays ..
      DOUBLE PRECISION AP((NMAX*(NMAX+1))/2), DUMMY(1,1), W(NMAX),
+      WORK(3*NMAX)
*      .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL          X02AJF
*      .. External Subroutines ..
      EXTERNAL          DSPEV
*      .. Intrinsic Functions ..
      INTRINSIC          ABS, MAX
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08GAF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*      Read the upper or lower triangular part of the matrix A from
*      data file
*
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((AP(I+(J*(J-1))/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((AP(I+((2*N-J)*(J-1))/2),J=1,I),I=1,N)
```

```

      END IF
*
*      Solve the symmetric eigenvalue problem
*
      CALL DSPEV('No vectors',UPLO,N,AP,W,DUMMY,1,WORK,INFO)
*
      IF (INFO.EQ.0) THEN
*
*          Print solution
*
          WRITE (NOUT,*) 'Eigenvalues'
          WRITE (NOUT,99999) (W(J),J=1,N)
*
*          Get the machine precision, EPS and compute the approximate
*          error bound for the computed eigenvalues. Note that for
*          the 2-norm, max( abs(W(i)) ) = norm(A), and since the
*          eigenvalues are returned in ascending order
*          max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)) )
*
          EPS = X02AJF()
          EERRBD = EPS*MAX(ABS(W(1)),ABS(W(N)))
*
*          Print the approximate error bound for the eigenvalues
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Error estimate for the eigenvalues'
          WRITE (NOUT,99998) EERRBD
        ELSE
          WRITE (NOUT,99997) 'Failure in DSPEV. INFO =', INFO
        END IF
      ELSE
        WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
      END

```

9.2 Program Data

F08GAF Example Program Data

```

4                               :Value of N

1.0  2.0  3.0  4.0
      2.0  3.0  4.0
          3.0  4.0
              4.0 :End of matrix A

```

9.3 Program Results

F08GAF Example Program Results

```

Eigenvalues
-2.0531 -0.5146 -0.2943 12.8621

Error estimate for the eigenvalues
1.4E-15

```
